
ReflectOS Kernel

  About


ReflectOS is the approachable, configurable, and extensible OS for your smart mirror project.  It is designed to allow anyone to easily install, customize, and enjoy a smart mirror/display - no coding or command line usage required!  Instructions for installing on your device and more details about ReflectOS can be found on the ReflectOS Firmware project.
This package provides the foundation for developers in the community to extend the OS by building their own sections, layouts, and layout managers.
Warning!
This project is being actively developed, and the API is not currently stable.
That said, would love feedback from those using it to build their own ReflectOS
extensions on how it can be improved and further stabilized.  Please feel free
to open an issue on the 
library GitHub with an suggestions.


  
    
  
  Extending ReflectOS



  
    
  
  Getting Started


ReflectOS allows developers to extend it's functionality by creating elixir libraries which register the extension with the system.  New sections, layouts, and layout managers can all be added to the OS via extension libraries (which are just standard elixir packages containing new ReflectOS modules).
The best way to start building your own extension library is to create 
a standard new elixir project using the mix new command, for example:
$ mix new reflect_os_myextensions --sup

Note that it is a good idea to use the --sup option to generate the 
application calback, this is a great place to register the modules you 
build with the ReflectOS system (more on that later).
Then, bring this package to your project by adding reflect_os_kernel to your list of dependencies in mix.exs:
def deps do
  [
    {:reflect_os_kernel, "~> 0.10.0"}
  ]
end

  
    
  
  Creating an Extension


Creating an extension is essentially just creating a module in your library which implements the proper elixir behavior.  Most extensions are comprised of a set of callbacks which describe both the runtime behavior and configuration experience through the ReflectOS Console.  
See the documentation for ReflectOS.Kernel.Section, ReflectOS.Kernel.Layout, and ReflectOS.Kernel.LayoutManager  for more details on how to get started.
Note that while Layout Managers are built on top of GenServer, Sections and Layouts are built on top of Scenic.Scene from the Scenic UI Framework.  If you are building your own Section or Layout, it is highly recommended you familiarize yourself with Scenic before getting started.

  
    
  
  Registering


Once created, extensions modules must be registered with ReflectOS in order to be available in the console web application which allows users to manage their displays.  Best practice is to do so via a supervised task during your application start up (this is why we recommend creating new projects with the --sup option).
For example:
defmodule MyReflectOSExtensions.Application do
  @moduledoc false

  use Application

  alias ReflectOS.Kernel.Section.Registry, as: SectionRegistry
  alias ReflectOS.Kernel.Layout.Registry, as: LayoutRegistry
  alias ReflectOS.Kernel.LayoutManager.Registry, as: LayoutManagerRegistry

  @impl true
  def start(_type, _args) do
    children = [
      # Start a task to register ReflectOS extensions
      {Task, fn -> reflect_os_register() end}
    ]

    opts = [strategy: :one_for_one, name: MyReflectOSExtensions.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp reflect_os_register() do
    # Sections
    SectionRegistry.register([
      MyReflectOSExtensions.Sections.MyNewSection,
    ])

    # Layouts
    LayoutRegistry.register([
      MyReflectOSExtensions.Layouts.MyNewLayout
    ])

    # Layout Managers
    LayoutManagerRegistry.register([
      MyReflectOSExtensions.Layouts.MyNewLayoutManager
    ])
  end
end

  
    
  
  Adding Extension Libraries to ReflectOS


To test your library and create a custom ReflectOS firmware image which includes your extensions, clone the latest release of the ReflectOS Firmware.  Be sure to checkout a release (e.g. use the -b flag), as the main branch may not be stable.  For example:
$ git clone https://github.com/reflect-os/firmware.git my_custom_firmware -b v0.9.0

Be sure to use the version number of the most recent release!
From there, it's as simple as adding your library as a dependency to the firmware's mix.exs file!  We recommend starting using the path option to refer to a local directory during development (see Mix.Tasks.Deps for more information on the various way to reference a dependency).
Once you've added your extension library as a dependency, follow the getting started instructions in the Firmware's Readme to see your project in action!

  
    
  
  Examples


For examples of how to build your own extensions to ReflectOS, check out the ReflectOS Core project.  These are extensions which are shipped with the pre-built ReflectOS firmware.

  
    
  
  Contributing


Contributions are welcome for this project!  You can 
open an issue to report a bug or request
a feature enhancement.  Code contributions are also welcomed, and can be 
submitted by forking this repository and creating a pull request.  

  
    
  
  Learn More


Full documentation can found on HexDocs.


  

    
ReflectOS.Kernel.Section behaviour
    



      
Sections are the widgets which appear your ReflectOS screen, showing information to user.

  
    
  
  Overview


Sections are a type of Scenic.Scene (which in turn is a type of GenServer) which renders a widget on the users ReflectOS Dashboard.  Like all extensions to ReflectOS, developers are also required to implement a set of callbacks which are used to allow run time configuration.  If you plan on publishing your extensions for use by others in the community, you can use these callbacks to create a thoughtful and intuitive configuration experience.
For the runtime UI behavior, Sections follow many of the same paradigms as Scenic.Scene and therefore should be very familiar to for developers experienced building native user interfaces with the Scenic Framework.  With Scenic, the UI is rendered natively (e.g. not via a webview) - this allows much better performance on the devices typically used for smart mirror projects (i.e. devices with a low-profile form factor and reduced processing power such as the Raspberry Pi Zero W and Zero 2 W).
This documentation will cover the basics, but it is highly recommended you refer to Scenic documentation for more details on using the framework to building native user interfaces.

  
    
  
  Implementing a Section


Sections are just modules which use ReflectOS.Kernel.Section.  The callbacks and other requirements for a Section fall into two major categories:
	Configuration experience via the ReflectOS Console web ui.
	Runtime native display rendering on the smart mirror/display


  
    
  
  Configuration Experience


In order to drive the ReflectOS console UI, sections are required to contain an Ecto.Schema which represents the available configuration options.  This is typically done using the embedded_schema/1 macro, as they are not persisted via Ecto.Repo.
Note that embedding schemas in your root schema (e.g. Ecto.Schema.embeds_many/4) is not currently supported.
Additionally, sections must implement the following callbacks which are used by the console:
	section_definition/0
	section_options/0
	changeset/2


  
    
  
  Runtime native display


In order to render the section to the ReflectOS dashboard on the device screen,
sections are only required to implement a single callback: init_section/3.
Sections can also optionally implement the validate_section/1 and handle_config_update/2 callbacks, which define section runtime functionality.

  
    
  
  By Example


See below for an example Section which renders a simple timer on the user's
smart mirror/display:
defmodule MyReflectOSExtensions.Sections.Timer do
  # Note that section passes along any options to the underlying `Scenic.Scene`
  use ReflectOS.Kernel.Section, has_children: false

  alias __MODULE__

  alias Scenic.Graph
  import Scenic.Primitives, only: [{:text, 3}]

  import Phoenix.Component, only: [sigil_H: 2]

  import ReflectOS.Kernel.Typography
  alias ReflectOS.Kernel.Section.Definition
  alias ReflectOS.Kernel.{OptionGroup, Option}
  import ReflectOS.Kernel.Primitives, only: [render_section_label: 2]

  #####################################################
  # Section Configuration
  #####################################################

  # This schema represents the avaible configuration options
  # for this section.
  embedded_schema do
    field(:show_label?, :boolean, default: false)
    field(:label, :string)
    field(:timer_seconds, :integer)
  end

  # This changeset will be called by the Console UI as users
  # create and update instances of the section.  Since the schema
  # above is what will be passed to your section at runtime, this is
  # a great opportunity to ensure the configuration provided will be valid.
  @impl true
  def changeset(%Timer{} = section, params \ %{}) do
    # Standard `Ecto.Changeset`, we can cast and validate input as usual.
    section
    |> cast(params, [:show_label?, :label, :timer_seconds])
    |> validate_required([:timer_seconds])
    |> validate_number(:timer_seconds, greater_than: 0)
  end

  # This is also used by the ConsoleUI.  It provides the name, icon, and
  # description of the section. See `ReflectOS.Kernel.Section.Definition`
  # for more details.
  @impl true
  def section_definition(),
    do: %Definition{
      name: "Timer",
      icon: """
        <svg xmlns="http://www.w3.org/2000/svg" viewBox="0 0 448 512">
        <!--!Font Awesome Free 6.6.0 by @fontawesome - https://fontawesome.com License - https://fontawesome.com/license/free Copyright 2024 Fonticons, Inc.-->
        <path d="M176 0c-17.7 0-32 14.3-32 32s14.3 32 32 32l16 0 0 34.4C92.3 113.8 16 200 16 304c0 114.9 93.1 208 208 208s208-93.1 208-208c0-41.8-12.3-80.7-33.5-113.2l24.1-24.1c12.5-12.5 12.5-32.8 0-45.3s-32.8-12.5-45.3 0L355.7 143c-28.1-23-62.2-38.8-99.7-44.6L256 64l16 0c17.7 0 32-14.3 32-32s-14.3-32-32-32L224 0 176 0zm72 192l0 128c0 13.3-10.7 24-24 24s-24-10.7-24-24l0-128c0-13.3 10.7-24 24-24s24 10.7 24 24z"/>
      </svg>
    """,
    description: fn assigns ->
      ~H"""
      This section allows you to configure a timer for your ReflectOS dashboard, which will count down to zero.
      <br />
      <br />
      To start a new time after it reaches zero, simply update the sections configuration with a new timer.
      """
    end
  }

  # These are the options which will be displayed to the user in the
  # section configuration form in the Console UI.  Note that the `key`
  # property must match the field in your Ecto schema.
  # See the `ReflectOS.Kernel.Option` documentation for more details.
  @impl true
  def section_options(),
    do: [
      # Option groups allow you to visually group fields together
      # in the Console UI.  They have no effect on the structure of the
      # the data in your schema.
      %OptionGroup{
        label: "Label",
        options: [
          %Option{
            key: :show_label?,
            label: "Show Label",
            config: %{
              type: "checkbox"
            }
          },
          %Option{
            key: :label,
            label: "Label Text",
            hidden: fn %{show_label?: show_label?} ->
              !show_label?
            end
          }
        ]
      },
      # Timer
      %Option{
        key: :timer_seconds,
        label: "Timer (seconds)",
        config: %{
          type: "tel"
        }
      }
    ]

  #####################################################
  # Section Runtime
  #####################################################

  @impl true
  def init_section(scene, %Timer{} = section_config, _opts) do
    # Here we simply store all of the fields in the section's configuration
    # struct (see the embedded_schema above) in the Scene's `assigns`
    # and then render the UI.
    scene =
      scene
      |> assign(Map.from_struct(section_config))
      |> render()

    # As mentioned above, Scene's are Genservers, so we can send messages just like we typically would.
    Process.send_after(self(), :tick_timer, 1_000)

    # `init_section/3` supports all the returns types supported by `Scenic.Scene.init/3`
    {:ok, scene}
  end

  @impl GenServer
  def handle_info(:tick_timer, scene) do
    # Gets the `:timer_seconds` from the `Scene.assigns`
    # (which we set from the config in the `init_section` callback) and subtracts 1
    timer_seconds = get(scene, :timer_seconds) - 1

    # Assign the updated timer and re-render the scene
    scene =
      scene
      |> assign(:timer_seconds, timer_seconds)
      |> render()

    # We only want to tick the timer until it reaches zero
    if timer_seconds > 0 do
      Process.send_after(self(), :tick_timer, 1_000)
    end

    # return from the `handle_info/2` like normal.
    {:noreply, scene}
  end

  defp render(%Scenic.Scene{} = scene) do
    # Here we render the graph to either show the remaining time or
    # that the timer has reached zero.  The `h2()` and `bold()` functions
    # are helpers provided by `ReflectOS.Kernel.Typography` to help create
    # a consistent look and feel.
    graph =
      case scene.assigns[:timer_seconds] do
        timer_seconds when timer_seconds > 0 ->
          Graph.build()
          |> text(
            "Timer will go off in #{timer_seconds} seconds.",
            h2()
            |> bold()
          )

        _ ->
          Graph.build()
          |> text(
            "Timer has completed!",
            h2()
            |> bold()
          )
      end

    label_config =
      scene.assigns
      |> Map.take([:show_label?, :label])

    # Use one of the `ReflectOS.Kernel.Components` to render a label
    # This MUST be called after the graph is complete.
    graph =
      render_section_label(graph, label_config)

    # Note that we use `push_section/2` instead of the typical `Scenic.Scene.push_graph/3`
    push_section(scene, graph)
  end
end
For more detailed examples of ReflectOS Sections, see the ReflectOS Core sections, which are shipped with the pre-built system firmware.

  
    
  
  Look and Feel Consistency


As you saw in the example above, one of the goals of ReflectOS is to create a cohesive visual experience for users - that is, there is styling consistency across all the sections displayed at any given time.  One of the goals of ReflectOS is to be the one screen in your home that is not constantly trying to get your attention - it should blend into it's environment, but be available for information when needed.
If you plan on publishing your extension for others to use, please consider following some basic styling guidelines:
	Use black, white, and grayscale colors only.  Since many users will be using your section behind a one-way mirror, white text on a black background generally provides the best contrast (this is the default).  Colors are more easily washed out by the reflection of the surrounding environment.

	Use font size and weight (e.g. bold, normal, light) to establish an information hierarchy.  Think about the most important information in your section, which you'd like to be available "at a glance", and use large font sizes or bold text or both make it more prominent.  For information with lower priority, use smaller font sizes or light text or both - not everything generall needs to be read from across the room!  Particuarly with a fixed screen size (no scrolling), space is at a premium and using smaller, more compact text wherever possible helps conserve space.

	Sections define their own width and height.  Native UIs using Scenic don't have the same flexible layout tools (e.g. flexbox) as HTML.  Thus, sections are responsible for establishing their own width - this means wrapping their own text as needed (see FontMetrics.wrap/5), properly sizing images, etc.



  
    
  
  Tips and Tricks


	Provide as many defaults as possible in your embedded_schema.  Ease of use is a first-order concern for ReflectOS, so providing reasonable defaults wherever possible is a great way to make it quick an easy for users to get started with your section while also providing them with a high level of customization.

	Use push_section/2 instead of push_graph.  To render a Scenic.Graph within a Scenic.Scene, one would typically use the Scenic.Scene.push_graph/2 function.  However, Sections must use push_section/2 instead to ensure proper positioning within the Layout.

	To aid with styling, use the helpers provide in this library.  For more details, see ReflectOS.Kernel.Typography and ReflectOS.Kernel.Components.

	Use the config field on ReflectOS.Kernel.Option.  This can be used to pass properties directly to the HTML input in the Console UI.  For example, you could use the placeholder property to show what the input might look like.

	You can subscribe to System Settings updates.  If your section uses values from the System Settings (such as the timezone), you can subscribe to changes so you can update the rendered section in realtime.  See ReflectOS.Kernel.Settings.System for more details.

Holds the section definition struct.

  
    
  
  Struct Fields


	name [required]: The name of the section type when it's
displayed in the Console UI.
	icon [required]: An SVG in raw string format which will be displayed
next to section in the Console UI.
	description [optional]: A short description of what the section does.
If provided, must be a single arity function an argument called assigns which returns a compiled HEEx component.  The best way to accomplish this is to
use the Phoenix.LiveView.sigil_H/2 macro.
	auto_align [optional, default: false]: A boolean indicating if the
ReflectOS layout should inject styles into the opts argument in the
section'sinit_section based on where the section appears in the layout.  For
example, if your section appears on the right side of the screen, the layout
might inject a text_align: right style.


  
    
  
  Example


Here is the Timer example from the ReflectOS.Kernel.Section documentation:
%Definition{
  name: "Timer",
  icon: """
    <svg xmlns="http://www.w3.org/2000/svg" viewBox="0 0 448 512">
      <!--!Font Awesome Free 6.6.0 by @fontawesome - https://fontawesome.com License - https://fontawesome.com/license/free Copyright 2024 Fonticons, Inc.-->
      <path d="M176 0c-17.7 0-32 14.3-32 32s14.3 32 32 32l16 0 0 34.4C92.3 113.8 16 200 16 304c0 114.9 93.1 208 208 208s208-93.1 208-208c0-41.8-12.3-80.7-33.5-113.2l24.1-24.1c12.5-12.5 12.5-32.8 0-45.3s-32.8-12.5-45.3 0L355.7 143c-28.1-23-62.2-38.8-99.7-44.6L256 64l16 0c17.7 0 32-14.3 32-32s-14.3-32-32-32L224 0 176 0zm72 192l0 128c0 13.3-10.7 24-24 24s-24-10.7-24-24l0-128c0-13.3 10.7-24 24-24s24 10.7 24 24z"/>
    </svg>
  """,
  description: fn assigns ->
    ~H"""
    This section allows you to configure a timer for your ReflectOS dashboard, which will count down to zero.
    <br />
    <br />
    To start a new time after it reaches zero, simply update the sections configuration with a new timer.
    """
  end
}

ReflectOS.Kernel.Section.Registry 
    



      
Used to register one or more ReflectOS.Kernel.Sections with the system.
If you are building an extension library with a new section type,
the section module must be registered with the ReflectOS system.
This will ensure that it's available for users to create and configure
sections via the Console UI application.
The recommended way to accomplish this is to do so by starting a task
in the root supervisor of your application.
For example:
defmodule MyReflectOSExtensions.Application do
  @moduledoc false

  use Application

  alias ReflectOS.Kernel.Section.Registry, as: SectionRegistry

  @impl true
  def start(_type, _args) do
    children = [
      # Start a task to register ReflectOS extensions
      {Task, fn -> reflect_os_register() end}
    ]

    opts = [strategy: :one_for_one, name: MyReflectOSExtensions.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp reflect_os_register() do
    SectionRegistry.register([
      MyReflectOSExtensions.Sections.MyNewSection,
    ])
  end
end

Layouts define the arrangment of sections on the screen.

  
    
  
  Overview


Layouts are a type of Scenic.Scene (which in turn is a type of GenServer) which defines the arrangement of sections on the users ReflectOS Dashboard.  For example, the FourCorners layouts which ships with ReflectOS allows users to place sections in the top left, top right, bottom left, and bottom right corners of the dashboard.  Layouts are passed their configuration, the screen size, and the sections the user has selected to display and are expected to render those sections to the screen.
Like all extensions to ReflectOS, developers are also required to implement a set of callbacks which are used to allow run time configuration.  If you plan on publishing your extensions for use by others in the community, you can use these callbacks to create a thoughtful and intuitive configuration experience.
For the runtime UI behavior, Layouts follow many of the same paradigms as Scenic.Scene and therefore should be very familiar to for developers experienced building native user interfaces with the Scenic Framework.  With Scenic, the UI is rendered natively (e.g. not via a webview) - this allows much better performance on the devices typically used for smart mirror projects (i.e. devices with a low-profile form factor and reduced processing power such as the Raspberry Pi Zero W and Zero 2 W).
This means that Layouts take on an important responsibility in the ReflecOS
system since the typical tools available in webviews (e.g. flexbox) are not
available.

  
    
  
  Layout Locations


Layouts define a list of available "locations" in their ReflectOS.Kernel.Layout.Definition, which are essentially areas of the screen where users can place one or more sections.  Layouts are responsible for adding each section to the layout's graph and using Scenic.Primitive.Transform.Translate to ensure the section is located in the right place on the screen.

  
    
  
  Rendering Sections


Since a ReflectOS.Kernel.Section is just a wrapped Scenic.Component,
Layouts can call Scenic.Component.add_to_graph/3 to render them to the
the layouts graph.  For example, your layout might contain the following
function:
def render_section(%Scenic.Graph{} = graph, %Section{} = section, layout_tracker, x, y) do
  %{id: section_id, module: section_module} = section

  graph
  |> section_module.add_to_graph({layout_tracker, section_id}, t: {x, y})
end
Note that the add_to_graph function must be called with a two-part tuple
in the format {layout_tracker, section_id} as the second argument, where layout_tracker is a unique id assigned by the layout used to identify the
section and section_id is the id of the section being rendered.
Note that you should avoid using the section_id as the layout tracker,
since users are permitted to add the same section multiple times to the same
layout.

  
    
  
  Implementing a Layout


Layouts are just modules which use ReflectOS.Kernel.Layout.  The callbacks and other requirements for a Layout fall into two major categories:
	Configuration experience via the ReflectOS Console web ui.
	Runtime native display rendering on the smart mirror/display


  
    
  
  Configuration Experience


In order to drive the ReflectOS console UI, layouts are required to contain an Ecto.Schema which represents the available configuration options.  This is typically done using the embedded_schema/1 macro, as they are not persisted via Ecto.Repo.
Note that embedding schemas in your root schema (e.g. Ecto.Schema.embeds_many/4) is not currently supported.
Additionally, layouts must implement the following callbacks which are used by the console:
	layout_definition/0
	layout_options/0
	changeset/2


  
    
  
  Runtime native display


In order to render the layout ReflectOS dashboard, modules are required to implement the following callbacks:
	init_layout/3
	handle_section_update/3

Layouts can also optionally implement the following callbacks:
	validate_layout/1
	handle_config_update/2
	handle_sections_update/2
	handle_viewport_update/2

See the documentation for each callback below for more details.

  
    
  
  Example


For a complete example of a ReflectOS Layout, see the FourCorner layout
from ReflectOS Core, which is shipped with the pre-built system firmware.

ReflectOS.Kernel.Layout.Definition 
    



      
Holds the layout definition struct.

  
    
  
  Struct Fields


	name [required]: The name of the layout type when it's
displayed in the Console UI.
	icon [required]: An SVG in raw string format which will be displayed
next to layout in the Console UI.
	description [optional]: A short description of the layout. If provided,
must be a single arity function an argument called assigns which returns
a compiled HEEx component.  The best way to accomplish this is to
use the Phoenix.LiveView.sigil_H/2 macro.
	locations [required]: A list of locations in the format
%{key: :my_location, label: "MyLocation"} indicating where users can place
sections in your layout.


  
    
  
  Example


Here an example adapted from the FourCorner layout in ReflectOS Core:
%Definition{
  name: "Four Corner",
  description: fn assigns ->
    ~H"""
    Allows placing sections in each of the four corners of the screen, with options for
    stacking (vertical vs. horizontal) and spacing.
    """
  end,
  icon: """
    <svg class="text-gray-800 dark:text-white" aria-hidden="true" xmlns="http://www.w3.org/2000/svg" width="24" height="24" fill="currentColor" viewBox="0 0 24 24">
      <path fill-rule="evenodd" d="M4.857 3A1.857 1.857 0 0 0 3 4.857v4.286C3 10.169 3.831 11 4.857 11h4.286A1.857 1.857 0 0 0 11 9.143V4.857A1.857 1.857 0 0 0 9.143 3H4.857Zm10 0A1.857 1.857 0 0 0 13 4.857v4.286c0 1.026.831 1.857 1.857 1.857h4.286A1.857 1.857 0 0 0 21 9.143V4.857A1.857 1.857 0 0 0 19.143 3h-4.286Zm-10 10A1.857 1.857 0 0 0 3 14.857v4.286C3 20.169 3.831 21 4.857 21h4.286A1.857 1.857 0 0 0 11 19.143v-4.286A1.857 1.857 0 0 0 9.143 13H4.857Zm10 0A1.857 1.857 0 0 0 13 14.857v4.286c0 1.026.831 1.857 1.857 1.857h4.286A1.857 1.857 0 0 0 21 19.143v-4.286A1.857 1.857 0 0 0 19.143 13h-4.286Z" clip-rule="evenodd"/>
    </svg>
  """,
  locations: [
    %{key: :top_left, label: "Top Left"},
    %{key: :top_right, label: "Top Right"},
    %{key: :bottom_left, label: "Bottom Left"},
    %{key: :bottom_right, label: "Bottom Right"}
  ]
}

ReflectOS.Kernel.Layout.Registry 
    



      
Used to register one or more ReflectOS.Kernel.Layout with the system.
If you are building an extension library with a new layout type,
the layout module must be registered with the ReflectOS system.
This will ensure that it's available for users to create and configure
layouts via the Console UI application.
The recommended way to accomplish this is to do so by starting a task
in the root supervisor of your application.
For example:
defmodule MyReflectOSExtensions.Application do
  @moduledoc false

  use Application

  alias ReflectOS.Kernel.Layout.Registry, as: LayoutRegistry

  @impl true
  def start(_type, _args) do
    children = [
      # Start a task to register ReflectOS extensions
      {Task, fn -> reflect_os_register() end}
    ]

    opts = [strategy: :one_for_one, name: MyReflectOSExtensions.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp reflect_os_register() do
    LayoutRegistry.register([
      MyReflectOSExtensions.Layouts.MyNewLayout,
    ])
  end
end

Layout Managers are responsible for determining which layout should be
rendered to the dashboard.

  
    
  
  Overview


Layout Managers are specialized GenServers which control which layout should be shown on the dashboard at any given time.  Layout Managers are
a unique feature of ReflectOS, allowing developers to build managers
which can change the layout in response to just about anything (e.g. a
fixed schedule, weather conditions, etc).  Using Layout Managers, users
can ensure their ReflectOS dashhboard shows just what they need, when
they need it.
Layouts Managers follow the same pattern of Scenic.Scene, and utilize
a dedicated struct (see ReflectOS.Kernel.LayoutManager.State) with an
assigns property as their state.  You can use the assigns property to
hold any data in state that you need.  Layout Managers are passed their
configuration as an init argument.
There are few requirements for LayoutManagers, which allows a high degree
of flexibility for building custom logic around which layout is displayed.
LayoutManagers simple use the push_layout/2 function to notify the
ReflectOS system that a new layout should be displayed.
Like all extensions to ReflectOS, developers are also required to implement a set of callbacks which are used to allow run time configuration.  If you plan on publishing your extensions for use by others in the community, you can use these callbacks to create a thoughtful and intuitive configuration experience.

  
    
  
  Implementing a Layout Manager


Layouts are just modules which use ReflectOS.Kernel.LayoutManager and implement it's behavior.  The callbacks and other requirements for a Layout fall into two major categories:
	Configuration experience via the ReflectOS Console web ui.
	Managing the logic around which layout should be displayed


  
    
  
  Configuration Experience


In order to drive the ReflectOS console UI, layout managers are required to contain an Ecto.Schema which represents the available configuration options.  This is typically done using the embedded_schema/1 macro, as they are not persisted via Ecto.Repo.
Note that embedding schemas in your root schema (e.g. Ecto.Schema.embeds_many/4) is not currently supported.
Additionally, layouts must implement the following callbacks which are used by the console:
	layout_manager_definition/0
	layout_manager_options/0
	changeset/2


  
    
  
  Layout Display Logic


In order to determine which layout renders to the ReflectOS dashboard, modules are on required to implement a single callback:
	init_layout_manager/2

Sections can also optionally implement the following callbacks:
	handle_config_update/2

See the documentation for each callback below for more details.

  
    
  
  Example


Here is the Static layout manager which ships with the pre-built
ReflectOS firmware.  It simply allows the user the select a single
layout via the console UI and sets it as the active layout.
defmodule ReflectOS.Core.LayoutManagers.Static do
  use ReflectOS.Kernel.LayoutManager

  import Phoenix.Component, only: [sigil_H: 2]

  alias ReflectOS.Kernel.Option
  alias ReflectOS.Kernel.LayoutManager
  alias ReflectOS.Kernel.LayoutManager.Definition
  alias ReflectOS.Kernel.LayoutManager.State
  alias ReflectOS.Kernel.Settings.LayoutStore

  @impl true
  def layout_manager_definition() do
    %Definition{
      name: "Static Layout",
      icon: """
        <svg class="text-gray-800 dark:text-white" aria-hidden="true" xmlns="http://www.w3.org/2000/svg" width="24" height="24" fill="none" viewBox="0 0 24 24">
          <path stroke="currentColor" stroke-linecap="round" stroke-linejoin="round" stroke-width="2" d="M7.757 12h8.486M21 12a9 9 0 1 1-18 0 9 9 0 0 1 18 0Z"/>
        </svg>
      """,
      description: fn assigns ->
        ~H"""
        Allows selecting a single, static layout.
        """
      end
    }
  end

  embedded_schema do
    field(:layout, :string)
  end

  @impl true
  def changeset(%__MODULE__{} = layout_manager, params \ %{}) do
    layout_manager
    |> cast(params, [:layout])
    |> validate_required([:layout])
  end

  @impl true
  def layout_manager_options(),
    do: [
      %Option{
        key: :layout,
        label: "Layout",
        config: %{
          type: "select",
          prompt: "--Select Layout--",
          options:
            LayoutStore.list()
            |> Enum.map(fn %{name: name, id: id} ->
              {name, id}
            end)
        }
      }
    ]

  @impl LayoutManager
  def init_layout_manager(%State{} = state, %{layout: layout_id}) do
    state =
      state
      |> push_layout(layout_id)

    {:ok, state}
  end
end

Holds the layout manager definition struct.

  
    
  
  Struct Fields


	name [required]: The name of the layout manager type when it's
displayed in the Console UI.
	icon [required]: An SVG in raw string format which will be displayed
next to layout managers in the Console UI.
	description [optional]: A short description of what the section does.
If provided, must be a single arity function an argument called assigns which returns a compiled HEEx component.
The best way to accomplish this is to use the Phoenix.LiveView.sigil_H/2 macro.


  
    
  
  Example


Here is the Static example from the ReflectOS.Kernel.LayoutManger documentation:
%Definition{
  name: "Static Layout",
  icon: """
    <svg class="text-gray-800 dark:text-white" aria-hidden="true" xmlns="http://www.w3.org/2000/svg" width="24" height="24" fill="none" viewBox="0 0 24 24">
      <path stroke="currentColor" stroke-linecap="round" stroke-linejoin="round" stroke-width="2" d="M7.757 12h8.486M21 12a9 9 0 1 1-18 0 9 9 0 0 1 18 0Z"/>
    </svg>
  """,
  description: fn assigns ->
    ~H"""
    Allows selecting a single, static layout.
    """
  end
}

Used to register one or more ReflectOS.Kernel.LayoutManagers with the system.
If you are building an extension library with a new layout manager type,
the layout manager module must be registered with the ReflectOS system.
This will ensure that it's available for users to create and configure
layout managers via the Console UI application.
The recommended way to accomplish this is to do so by starting a task
in the root supervisor of your application.
For example:
defmodule MyReflectOSExtensions.Application do
  @moduledoc false

  use Application

  alias ReflectOS.Kernel.LayoutManager.Registry, as: LayoutManagerRegistry

  @impl true
  def start(_type, _args) do
    children = [
      # Start a task to register ReflectOS extensions
      {Task, fn -> reflect_os_register() end}
    ]

    opts = [strategy: :one_for_one, name: MyReflectOSExtensions.Supervisor]
    Supervisor.start_link(children, opts)
  end

  defp reflect_os_register() do
    LayoutManagerRegistry.register([
      MyReflectOSExtensions.LayoutManagers.MyNewLayoutManager,
    ])
  end
end

ReflectOS.Kernel.LayoutManager.State 
    



      
Holds the struct which is used as the ReflectOS.Kernel.LayoutManager state.

Defines the struct used by sections, layouts, and layout managers to
present configuration options to users in the ReflectOS Console UI.

  
    
  
  Struct Fields


	key [required]: The key of the field in the configuration schema.
	label [required]: Label used for the HTML input in the Console UI.
	hidden [optional]: Function which determines whether or not this
input should be displayed.  Takes a single argument, which is a map of
configuration values as the user fills out the configuration form.
	config [required]: A map which is passed directly to the HTML input
in the Console UI.


  
    
  
  Examples



  
    
  
  Simple Text Input


This example shows a minimal option.
%Option{
  key: :rss_feed,
  label: "RSS Feed Url"
}

  
    
  
  Select Input


This example shows a select input - note that the list of options is
populated at runtime, not at compile time.  This provides quite a bit of
flexibility.
%Option{
  key: :time_format,
  label: "Time Format",
  config: %{
    type: "select",
    options: [
      {"12 Hour (6:30 PM)", "%-I:%M %p"},
      {"24 Hour (18:30)", "%-H:%M"},
      {"System Default [#{case System.time_format() do
        "%-I:%M %p" -> "12 Hour (6:30 PM)"
        "%-H:%M" -> "24 Hour (18:30)"
        _ -> ""
        end}]", "system"}
    ]
  }
}

  
    
  
  Input with Help Text


This example shows how to display help text, which is displayed under
the input to provide the user more information on how to fill in the field.
Note that help_text takes a single arity function which MUST take an argument called
assigns and return the result of a Phoenix.Component.sigil_H/2.
The ReflectOS Console uses the
Flowbite CSS Library
for styling, so feel free to use any of the css classes that it provides!
%Option{
  key: :rss_feed,
  label: "RSS Feed Url",
  help_text: fn assigns ->
    ~H"""
    Web address of the RSS feed.  Some popular examples can be found
    <a
      class="font-medium text-blue-600 dark:text-blue-500 hover:underline"
      href="https://github.com/plenaryapp/awesome-rss-feeds?tab=readme-ov-file#-United-States"
      target="_blank">
      here
    </a>.
    """
  end
}

  
    
  
  Hiding Inputs


The hidden field allows you specify when the input should be hidden
based on the current values the user has provided in the configuration
form.
Consider this common example where the text input for the label is only
shown if the user elects to show the label.
%Option{
  key: :show_label?,
  label: "Show Label",
  config: %{
    type: "checkbox"
  }
},
%Option{
  key: :label,
  label: "Label Text",
  hidden: fn %{show_label?: show_label?} ->
    !show_label?
  end
}

ReflectOS.Kernel.OptionGroup 
    



      
Creates a visual grouping of configuration option inputs in the Console UI.

  
    
  
  Struct Fields


	label [required]: Title of the input grouping.
	description [optional]: A single arity function which can be passed to Phoenix.LiveView.
	options [optional]: List of ReflectOS.Kernel.Option which should be
displayed in the input grouping.

Note that description takes a single arity function which MUST take an argument called
assigns and return the result of a Phoenix.Component.sigil_H/2.  This allows
you to use html markup in the description.  The ReflectOS Console uses the
Flowbite CSS Library
for styling, so feel free to use any of the css classes that it provides!

  
    
  
  Example


%OptionGroup{
  label: "Label",
  description: fn assigns ->
    ~H"""
    The label appears over the section, and can be optionally shown or hidden.
    For more information, see the documentation
    <a
      class="font-medium text-blue-600 dark:text-blue-500 hover:underline"
      href="https://example.com"
      target="_blank">
      here
    </a>.
    """
  options: [
    %Option{
      key: :show_label?,
      label: "Show Label",
      config: %{
        type: "checkbox"
      }
    },
    %Option{
      key: :label,
      label: "Label Text",
      hidden: fn %{show_label?: show_label?} ->
        !show_label?
      end
    }
  ]
}

ReflectOS.Kernel.Settings.System 
    



      
This module allows you to retrieve system settings, as well as
subscribe to changes.
A common pattern is to default section configuration to use a
system setting (e.g. time format), but allow users to override it
for a given section.
If the default system setting is used, your section can subscribe
to changes in the setting so the display can be updated in real-time.
System Setting uses the excellent PropertyTable library
under the hood, and the message sent to your section when you subscribe
are the standard PropertyTable.Event struct.
Here is an example using the time_format/0 function,
excerpted from the date/time section in
ReflectOS Core:
defmodule ReflectOS.Core.Sections.DateTime do
  use ReflectOS.Kernel.Section, has_children: false

  alias ReflectOS.Kernel.Settings.System
  alias ReflectOS.Kernel.Section.Definition
  alias ReflectOS.Kernel.{OptionGroup, Option}


  embedded_schema do
    # Default time format to the system setting
    field(:time_format, :string, default: "system")
  end

  def section_options(),
    do: [
      %Option{
        key: :time_format,
        label: "Hour Format",
        config: %{
          type: "select",
          options: [
            {"12 Hour (6:30 PM)", "%-I:%M %p"},
            {"24 Hour (18:30)", "%-H:%M"},
            {"System Default", "system"}
          ]
        }
      }
    ]

  def init_section(scene, %__MODULE__{} = section_config, opts) do
    time_format =
      if section_config.time_format == "system" do
        # If we are using the system setting,
        # subscribe for updates
        System.subscribe("time_format")
        System.time_format()
      else
        # otherwise, use the user's selection
        section_config.time_format
      end


    {:ok, assign(scene, time_format: time_format)}
  end

  # Handle event when the time_format system setting changes.
  # This will only get called if the select time format for
  # the section is the "System Default"
  def handle_info(
        %PropertyTable.Event{
          property: ["system", "time_format"],
          value: time_format
        },
        scene
      ) do
    scene =
      scene
      |> assign(time_format: time_format)

    {:noreply, render(scene)}
  end
end

ReflectOS.Kernel.GraphHelpers 
    



      
Helper functions for working with S